

## SIMULAZIONE DI PROCESSO DI "RUN-IN-HOLE"

### Descrizione del problema

La denominazione "run-in-hole" RIH è utilizzata per indicare una particolare fase nel processo di inserimento di una batteria di perforazione in un pozzo, susseguente, ad esempio, ad una estrazione finalizzata al cambio della composizione della batteria stessa.

Durante questa fase la batteria, inizialmente sospesa ad un cavo recante un gancio ("hoist") viene fatta scendere nel pozzo per una determinata lunghezza al fine di consentire l'inserimento di un nuovo tratto.

Durante questa fase, possono crearsi oscillazioni assiali di notevole ampiezza nella batteria, che possono dar luogo a fenomeni potenzialmente pericolosi, come il raggiungimento di tensioni che superano il limite di snervamento o di rottura del materiale, o la fuoriuscita improvvisa della batteria stessa dal pozzo ("kick-off") con sganciamento dal dispositivo di sostegno.

Il presente esempio applicativo si propone di studiare il transitorio dinamico che si verifica durante questa fase del processo di perforazione tramite un modello ad EF.




FIG: 1

### Composizione della Batteria

Una batteria di perforazione è tipicamente composta di due tipi di aste, le aste leggere (Drill Pipes, DP), nella parte superiore e le aste pesanti (Drill Collars, DC) nella parte inferiore.

Il fango presente all'interno del pozzo produce inoltre un considerevole effetto smorzante, quantificabile in circa 47 N-sec/m per ogni "set" di DP di 28 m di lunghezza e a circa 35 N-sec/m per ogni DC di 9.5 m di lunghezza.




Fig. 2

### Modello ad EF

Il modello ad EF proposto è basato sull'impiego di elementi trave nel piano e si propone di studiare esclusivamente le vibrazioni assiali della batteria. Per tale motivo, sono stati vincolati i g.d.l. non assiali di tutti i nodi.

Nel modello sono inoltre stati inseriti elementi rappresentanti smorzatori distribuiti lungo l'intera lunghezza della batteria, ed un elemento "tension-only", rappresentante il

cavo di sospensione dell'hoist.

Si sono inoltre trascurati i contributi della deformabilità del "derrick" (traliccio) di sostegno.

Il modello è schematicamente indicato in Fig. 2.

### **Processo di RIH**

Il processo di RIH si compone (Fig. 3) delle 5 fasi seguenti:

1. La batteria è ferma, appesa al cavo, sotto l'azione del suo peso proprio (punto O)
2. (Durata 1 sec.) La batteria viene tirata in alto ad una velocità di 0.3 m/sec per liberarla dalla presa delle ganasce a gravità (tratto OA)
3. (Durata 1 sec.) La batteria viene riportata nella posizione iniziale, dopo aver tolto le ganasce, ad una velocità di 0.3 m/sec (tratto AB)
4. (Durata variabile, tipicamente 10-12 sec.) La batteria viene fatta accelerare si suppone ad  $1g$ , fino al raggiungimento di una velocità costante, che viene mantenuta sino a quanto richiesto dall'inserimento del nuovo tratto
5. (Durata 1 sec.) La discesa della batteria viene frenata fino all'arresto tramite appositi dispositivi; vengono quindi nuovamente inserite le ganasce a gravità (tratto DE).




Fig. 3




Fig. 4

Al momento dell'arresto (punto E) deve risultare completamente inserito il nuovo tratto (lunghezza tipica 28 m).

Al termine del processo di inserimento la batteria continua a subire oscillazioni assiali libere, fino a che lo smorzamento non le arresta.

L'intero processo è stato simulato nel modello EF, con possibilità per l'utente di controllare il tempo complessivo delle fasi 4 e 5 (parametro RIH).

### ***Andamento tipico delle tensioni nella batteria***

La Fig. 4 riporta un tipico andamento nel tempo delle tensioni alla sommità della batteria, con evidenziazione del fenomeno del "kick-off", nel momento in cui le tensioni stesse divengono di compressione.

### ***File di ingresso***

Il file di ingresso di ANSYS che realizza il modello illustrato è riportato nel seguito.

Il modello si basa su comuni elementi trave 3D, cui vengono soppressi (vincolati) i gdl non necessari per l'analisi, vale a dire tutti, tranne lo spostamento lungo l'asse della batteria.

Per la simulazione del contatto unilaterale col gancio di sostegno (Hoist) viene utilizzato un elemento LINK10 nella formulazione "tension only".

Si provi a variare il tempo di RIH tra 12 e 6 secondi, verificando la possibile insorgenza di fenomeni di kick-off.

Infine, la parte terminale del file di ingresso, produce un grafico animato della distribuzione di tensione normale lungo l'asta durante il transitorio.

## FILE DI INGRESSO PER SIMULAZIONE DI PROCESSO DI "RUN-IN-HOLE"

c\*\*\* \*\*\*\*\*

C\*\*\* CORSO DI LAUREA MAGISTRALE IN ING. MECCANICA

C\*\*\*

C\*\*\* CORSO DI COSTRUZIONE DI MACCHINE

C\*\*\*

C\*\*\* MODELLO PER LO STUDIO DEL TRANSITORIO DINAMICO

C\*\*\* DI "RUN-IN-HOLE"

C\*\*\*

C\*\*\* modello dinamico per lo studio del pipe nel RIH

c\*\*\* versione definitiva analisi dinamica

c\*\*\* con geometria parametrica

c\*\*\* con precarico statico e smorzamento viscoso del MUD

c\*\*\* il pipe viene liberato dalle slips

c\*\*\* poi portato alla velocità di regime

c\*\*\* è simulato l'hoist con un solo elemento link10

c\*\*\* che resiste a trazione ma non a compressione

C\*\*\* FEATURES:

C\*\*\* USO DI SMORZATORI

C\*\*\* RAPPRESENTAZIONE DI STORIE DI CARICO COMPLESSE

C\*\*\* USO DI ELEMENTI LINK SPECIALI

C\*\*\* ANIMAZIONE DELLA DISTRIBUZIONE DI FORZA NORMALE

C\*\*\*

C\*\*\*

finish

/clear

/prep7

/title, Analisi RIH

\*ASK,RIH,Durata RIH (sec),12

\*ASK,PTIN,Passo temporale di integrazione (sec): ,0.0005

\*ASK,TIMPAS,Durata dei LStps :,0.5

C\*\*\* caratteristiche del materiale nel sistema mks

C\*\*\*

C\*\*\* ACCIAIO

mp,ex,1,210000e6 ! modulo elastico dell'acciaio grade E

MP,GXY,1,2.1E11/2.6

dens,1,7850 ! densità dell 'acciaio

mp,damp,1,0.0003 ! smorzamento interno del materiale

mp,ex,2,2.1e11

MP,GXY,2,2.1E11/2.6

dens,2,500

c\*\*\* INTRODUZIONE DEI PARAMETRI

L1=1530 ! Lunghezza del drill pipe (metri),

L2=114 ! Lunghezza del drill collar (metri)

L3A=28 ! Lunghezza del gruppo di aste da inserire (metri)

T1=1 ! durata sollevamento iniziale [s]

V1=0.3 ! velocità sollevamento iniziale [m/s]

T2=1 ! durata discesa iniziale [s]

V2=0.3 ! velocità discesainiziale [m/s]

TF=1 ! tempo di fenata finale [S]

Tof=6 ! tempo di oscillazione finale [s]  
DTIM=TIMPAS/100! tempo fittizio iniziale

NDP=108 ! N. ELEMENTI DRILL PIPES  
NDC=24 ! N. ELEMENTI DRILL COLLARS  
! IN TUTTO MENO DI 500 ELEMENTI

c\*\*\*\*\* smorzamento

!lo smorzamento standard è di 47 N sec/metro per ogni pipe di 28 metri

!e di 35 N sec/metro per ogni collar da 9.5 metri

!il mesh del pipe è di 108 elementi

!quello del collar di 24 elementi

C1A=47

C2A=35

$C1=(C1A/L3A)*(L1/NDP)$

$C2=(C2A/9.5)*(L2/NDC)$

C\*\*\* SCELTA DEGLI ELEMENTI

! DRILL STRING

et,1,4

! real1 per il tubo da 4.5 inch 16.6 lb/ft

! r,1,area,Jzz,height

r,1,2.83e-3,4e-6,0.144

! real 2 per il drill collar da 6.5 inch

! r,2,area,Jzz,height

r,2,0.0174,3.51e-5,0.165

! SMORZATORI

et,2,11

! lungo i drill pipe

r,3,0,C1

! lungo i drill collar

r,4,0,C2

!HOIST (CAVO DI SOSTEGNO CON GANCIO)

et,3,10,,0,0

r,5,1.6e-3,0

c\*\*\* CREAZIONE DEL MODELLO ( nodi)

! si considera l'asse y come asse del pipe

! le y negative sono rivolte verso la direzione di perforazione

C\*\*\*\*\*

C\*\*\* NODI

C\*\*\*\*\*

C\*\*\*

C\*\*\*\* ZONA DRILL PIPES

n,1

N,NDP+1,0,-L1

fill,1,NDP+1

C\*\*\*\* ZONA DRILL COLLARS

n,NDP+NDC+1,0,-(L1+L2)

fill,NDP+1,NDP+NDC+1

C\*\*\* SMORZATORI

n,501,,1

n,501+NDP,,-L1+1

fill,501,501+NDP

n, 501+NDP+NDC,,-(L1+L2)+1

fill,501+NDP,501+NDP+NDC

nde,501

C\*\*\*\*\*

C\*\*\* ELEMENTI

C\*\*\*\*\*

C\*\*\*

C\*\*\* ZONA DRILL PIPES

C\*\*\*

type,1

mat,1

real,1

e,1,2

egen,NDP,1,-1

C\*\*\* ZONA DRILL COLLARS

type,1

mat,1

real,2

e,109,110

egen,NDC,1,-1

C\*\*\* SMORZATORI

C\*\*\* zona drill pipes

type,2

mat, 1

real,3

e,502,2

egen,NDP,1,-1

C\*\*\* zona drill collars

real,4

e,502+NDP,NDP+2

egen,NDC,1,-1

c\*\*\* VINCOLI

C\*\*\* drill string

nset,s,node,,1,NDC+NDP+1,1

d,all,rotz,0

D,ALL,ROTX,0

D,ALL,ROTY,0

d,all,ux,0

d,all,uz,0

nset,all

c\*\*\* smorzatori

nset,s,node,,501,9999,1

d,all,all,0

nset,all

C\*\*\* SIMULAZIONE DELL'HOIST (CAVO DI SOSTEGNO CON GANCIO)

n,2000,0,40

type,3

C\*\*\*TYPE,1

real,5

mat,1

e,2000,1

d,2000,all,0

/pbc,all,1

eplot

finish

```

c*****
c*** calcolo del rih con start ( fase di 2 sec di rilascio dalle slips)
/prep7

c*** macro utili per gli steps di carico
*create,passo
d,2000,uy,arg2
kbc,0
time,arg1
autots,on
nsubst,2
cnvtol,u,,0.001,2
outres,all,all
outpr,all,all
solve
*end
*create,dopo
d,2000,uy,arg2
c***d,1,uy,arg3
kbc,1
time,arg1
autots,on
nsubst,2
cnvtol,u,,0.001,2
outres,all,all
outpr,all,all
solve
*end

C*****
C***
C*** ANALISI TRANSITORIO DINAMICO
C***
C*****
C*** STEP 0 - ASTA SOGGETTA AL SUO PESO PROPRIO
C***
/solu
antype,trans ! TIPO ANALISI : TRANSITORIO
trnpt,full ! METODO DI SOLUZIONE : FULL
timint,OFF ! INDICA CHE IL PRIMO STEP E' STATICO
acel,0,9.81 ! IMPONE UN'ACCELERAZIONE IN DIREZIONE Y PARI A G

Kbc,1
time,DTIM/100

kbc,1

cnvtol,u,,0.001,2 ! TOLLERANZA PER ANALISI NON LINEARI
outres,all,all ! OPZIONI OUTPUT
outpr,all, all
solve
time,2*DTIM/100
solve

KBC,0
nropt,full ! TECNICA RISOLUTIVA SE SONO RICHIESTE ANALISI NON LINEARI
timint, on, , on ! ATTIVA L'ANALISI DI TRANSITORIO
deltim, PTIN ! PASSO TEMPORALE DI INTEGRAZIONE
C*****
C*** STEP 1 - ASTA SOLLEVATA PER LIBERARLA DAGLI "SIP"
C***
*do,tempo,0,T1,TIMPAS

```

```

*IF,tempo,eq,0,then
    tempo=tempo+dtim
*ENDIF
sposta=V1*tempo
*use,passo,tempo,sposta
*enddo

```

```

C*****
C*** STEP 2 - ASTA RIABBASSATA FINO ALLA POSIZIONE INIZIALE
C***

```

```

*do,tempo,T1,T1+T2,TIMPAS
*IF,tempo,eq,T1,then
    tempo=tempo+dtim
*ENDIF
sposta=V1*T1-(V2*(tempo-T1))
*use,passo,tempo,sposta
*enddo

```

TACC=(RIH-(TF/2))-SQRT((RIH-(TF/2))\*\*2-2\*(L3A+V2\*(TF/2-RIH))/9.81)

```

*do,tempo,T1+T2,T1+T2+TACC,TIMPAS/10
*IF,tempo,eq,(T1+T2),then
    tempo=tempo+dtim
*ENDIF
sposta=-V2*(tempo-(T1+T2))-0.5*9.81*(tempo-(T1+T2))**2
*use,passo,tempo,sposta
*enddo

```

```

C*****
C*** STEP 3 - ASTA che scende a velocita' costante
C***

```

```

! il tempo a cui inizia la frenata è dato dal tempo
! per sbloccare le slips, più il tempo totale di RIH
! meno il tempo di frenata
! b=before
! a=after

```

T1b=RIH+(T1+T2)-(TF)

```

! velocità di RIH
!velRIH=L3A/(RIH-TF*0.5)
velRIH=V2+9.81*TACC
S0=-V2*TACC-0.5*9.81*TACC**2

```

```

*do,tempo,T1+T2+TACC,T1b,TIMPAS
*IF,tempo,eq,(T1+T2+TACC),then
    tempo=tempo+dtim
*ENDIF
sposta=S0-velRIH*(tempo-(T1+T2+TACC))
*use,passo,tempo,sposta
*enddo
T1a=T1b
T2b=T1b+TF
ACCF=VelRIH/TF

```

```

C*****
C*** STEP 4 - frenatura finale
C***

```

```

*do,tempo,T1a,T2b,TIMPAS

```

```

*IF,tempo,eq,T1a,then
    tempo=tempo+dtim
*ENDIF
sposta=S0-velRIH*(T1b-(T1+T2+TACC))-(velRIH*(tempo-T1b)-(1/2)*ACCF*(tempo-T1b)**2)
*use,passo,tempo, sposta
*enddo
C*****
C*** STEP 5 - asta serrata dopo RIH
C***
c***/solu
T2a=T2b
T3=T2b+Tof

*do,tempo,T2a,T3,TIMPAS
*IF,tempo,eq,T2a,then
    tempo=tempo+dtim
*ENDIF
*use,dopo,tempo, sposta, serro
*enddo
finish

C*****
C***
C*** ANDAMENTO GRANDEZZE PRINCIPALI IN F.NE DEL TEMPO
C***

/post26
RFORCE,2,2000,F,Y,RF20
nsol,3,2000,u,y,spostHoist
nsol,4,1,u,y,spost1
nsol,5,ndp+ndc+1,u,y,spostBit
store
derive,6,3,1,,velocity
/AXLAB,Y,SPOSTAMENTO SOMMITA'
PLVAR,3
*ASK,DD,RETURN PER PROSEGUIRE,0
/AXLAB,Y,VELOCITA' SOMMITA'
PLVAR,6
*ASK,DD,RETURN PER PROSEGUIRE,0
/AXLAB,Y,REAZIONE SOMMITA'
PLVAR,2
*ASK,DD,RETURN PER PROSEGUIRE,0
/AXLAB,Y,SPOSTAMENTI
PLVAR,5,3,4
*ASK,DD,RETURN PER PROSEGUIRE,0
FINISH

C*****
C***
C*** DIAGRAMMA FORZA NORMALE IN F.NE DEL TEMPO
C***
/POST1
SET,LAST
*GET,NSTS,ACTIVE,0,SET,NSET ! Estrae i numeri di LStep e Substep

VMX=-9999.
VMN=9999.

c*** estrae i valori massimi e minimi della forza normale durante tutto il transitorio
*DO,IJK,1,NSTS
    SET,NEXT
    ETABLE,NI,SMISC,1

```

```
ETABLE,NJ,SMISC,7
ESORT,ETAB,NI,
*GET,V1,SORT,0,MAX
ESORT,ETAB,NI,
*GET,V2,SORT,0,MIN
*IF,V1,GT,VMX,THEN
    VMX=V1
*ENDIF
*IF,V2,LT,VMN,THEN
    VMN=V2
*ENDIF
*ENDDO
```

c\*\*\* fissa una scala di rappresentazione dello sforzo normale valida per tutto il transitorio  
/CONTOUR,,VMN,VMX

```
C*** PRODUCE LA RAPPRESENTAZIONE ANIMATA
*DO,IJK,1,NSTS
    SET,NEXT
    ETABLE,NI,SMISC,1 ! GRAFICO DELLA FORZA NORMALE
    ETABLE,NJ,SMISC,7
    PLLS,NI,NJ
    /WAIT,0.1      ! PAUSA
*ENDDO
```